

GIỚI HẠN CỦA HÀM SỐ NGUYÊN HÀM VÀ TÍCH PHÂN

I. Giới hạn của hàm số

Trong Maple, có hai lệnh để tính và hiển thị kí hiệu giới hạn.

Dùng lệnh **> Limit (hàm số, x = x₀, left/right (trái/phải));** để hiển thị biểu thức giới hạn (bên trái/ phải) của hàm số tại x=x₀. {Chữ L trong từ khóa Limit là chữ in hoa}

Dùng lệnh **> limit(hàm số, x = x₀, left/right (trái/phải));** để hiển thị (tính) giá trị của giới hạn (bên trái/ phải) của hàm số tại x=x₀. {Chữ L trong từ khóa limit là chữ thường}.

Ví dụ 1: Tính giới hạn $\lim_{x \rightarrow 2} \frac{\sqrt{x-1}-1}{1+\sqrt[3]{1-x}}$.

Để hiển thị ngay kết quả chúng ta nhập vào Maple dòng lệnh như sau:

```
> limit((1-sqrt(x-1))/(1+surd(1-x,3)),x=2);
```

$$\frac{3}{2}$$

(Trong Maple, căn bậc n của số a được khai báo bởi từ khóa **> surd(a,n);**)

Nếu đưa vào giảng dạy, chúng ta cần cho hiển thị biểu thức giới hạn. Đầu tiên chúng ta nhập vào Maple lệnh sau:

```
> Limit((1-sqrt(x-1))/(1+surd(1-x,3)),x=2);
```

$$\lim_{x \rightarrow 2} \frac{1 - \sqrt{x-1}}{1 + \text{surd}(1-x, 3)}$$

Để hiển thị kết quả của giới hạn trên chúng ta chỉ cần dùng hàm **>value(%);** trên dòng lệnh liền kề sau đó:

```
> value(%) ;
```

$$\frac{3}{2}$$

Lưu ý: Nếu chỉ tính giới hạn của hàm số tại một điểm nào đó thì ta không cần khai báo thêm thông số left hoặc right.

Ví dụ 2: Tìm giới hạn bên trái và giới hạn bên phải và giới hạn bên phải của hàm số

$$y = \frac{|x-1|}{\sqrt{x^2+3}-2} \text{ tại } x = 1.$$

Để dễ quan sát, đầu tiên chúng ta nhập biểu thức của hàm số vào Maple:

```
> y:=abs(x-1)/(sqrt(x^2+3)-2);
```

$$y := \frac{|x-1|}{\sqrt{x^2+3}-2}$$

Để hiển thị biểu thức giới hạn bên trái của hàm số trên tại x = 1 ta dùng lệnh như sau:

```
> Limit(y,x=1,left);
```

$$\lim_{x \rightarrow 1^-} \frac{|x-1|}{\sqrt{x^2+3}-2}$$

Giá trị của giới hạn bên trái :

> **value(%)**;

-2

Để hiển thị biểu thức giới hạn bên phải của hàm số trên tại $x = 1$ ta dùng lệnh như sau:

> **Limit(y,x=1,right)**;

$$\lim_{x \rightarrow 1^+} \frac{|x-1|}{\sqrt{x^2+3}-2}$$

Giá trị của giới hạn bên phải :

> **value(%)**;

2

Từ hai kết quả trên chúng ta kết luận hàm số đã cho không có giới hạn tại $x=1$. \square

Nếu chúng ta nhập vào dòng lệnh để tìm giới hạn hàm số trên tại $x=1$ thì Maple sẽ thông báo hay cho ra kết quả như sau:

> **Limit(y,x=1)**;

$$\lim_{x \rightarrow 1} \frac{|x-1|}{\sqrt{x^2+3}-2}$$

> **value(%)**;

undefined

Kết quả Maple thông báo là “*undefined*” cho biết hàm số đã cho không có giới hạn tại $x=1$.

Ví dụ 3: Tính các giới hạn $\lim_{x \rightarrow \pm\infty} (x - \sqrt{x^2 + x + 1})$.

Để dễ dàng thao tác trên nhiều dòng lệnh, đầu tiên chúng ta nhập biểu thức của hàm số vào Maple:

> **y:=x-sqrt(x^2+x+1)**;

$$y := x - \sqrt{x^2 + x + 1}$$

Biểu thức giới hạn của hàm số khi $x \rightarrow -\infty$:

> **Limit(y,x=-infinity)**;

$$\lim_{x \rightarrow (-\infty)} x - \sqrt{x^2 + x + 1}$$

Giá trị của giới hạn trên:

> **value(%)**;

$-\infty$

Biểu thức giới hạn của hàm số khi $x \rightarrow +\infty$ và giá trị của giới hạn:

> **Limit(y,x=+infinity)**;

$$\lim_{x \rightarrow \infty} x - \sqrt{x^2 + x + 1}$$

> **value(%)**;

$-\frac{1}{2}$

☞ Đề ý: Khi $x \rightarrow +\infty$ thì Maple vẫn cho hiển thị là $x \rightarrow \infty$, khác với khi $x \rightarrow -\infty$.

Ví dụ 4: Tìm giới hạn: $\lim_{x \rightarrow 0} \frac{1}{x}$.

Dùng Maple để tính giới hạn trên ta được kết quả:

> **limit(1/x,x=0);**

undefined

Nhưng nếu tính các giới hạn $\lim_{x \rightarrow 0^+} \frac{1}{x}$ và $\lim_{x \rightarrow 0^-} \frac{1}{x}$ thì kết quả sẽ thế nào ?

Chúng ta dùng hai thủ tục sau (tính giới hạn phải/trái của hàm số “ham” tại $x=x_0$) để dễ quan sát và hiển thị gọn gàng hơn:

> **ghtrai:=proc(ham,x0)**

print(Limit(ham,x=x0,left)=limit(ham,x=x0,left))end proc:

> **ghphai:=proc(ham,x0)**

print(Limit(ham,x=x0,right)=limit(ham,x=x0,right))end proc:

Dùng thủ tục **>ghtrai(ham,x0)** ; ta tính giới hạn trái của hàm số tại $x=0$:

> **ghtrai(1/x,0);**

$$\lim_{x \rightarrow 0^-} \frac{1}{x} = -\infty$$

Dùng thủ tục **>ghphai(ham,x0)** ; ta tính giới hạn phải của hàm số tại $x=0$:

> **ghphai(1/x,0);**

$$\lim_{x \rightarrow 0^+} \frac{1}{x} = \infty$$

Như vậy chúng ta thấy, Maple đã giúp cho chúng ta phân biệt rõ hơn rằng: $\lim_{x \rightarrow 0} \frac{1}{x}$ không

xác định, nhưng $\lim_{x \rightarrow 0^+} \frac{1}{x} = +\infty$ và $\lim_{x \rightarrow 0^-} \frac{1}{x} = -\infty$. Điều này hoàn toàn phù hợp với yêu cầu đổi mới và nội dung của chương trình SGK phổ thông hiện nay.

Chú ý: Trên tập số thực thì $\lim_{x \rightarrow 0} \frac{1}{x}$ không xác định, nhưng trên tập số phức thì Maple sẽ thông báo với kết quả khác:

> **Limit(1/x,x=0,complex)=limit(1/x,x=0,complex);**

$$\lim_{x \rightarrow 0, \text{complex}} \frac{1}{x} = \infty - \infty I$$

Kết quả Maple hiển thị là $\infty - \infty I$ (biểu diễn dưới dạng số phức).

Một điều cần lưu ý là nếu trong hai lệnh trên chúng ta không khai báo thông số “complex” thì Maple sẽ mặc định tính giới hạn trên tập số thực.

Ví dụ 5: Giới hạn của hàm số cho bởi nhiều công thức

Cho hàm số $f(x) = \begin{cases} x^2 - 2x + 3 & \text{với } x \leq 2, \\ 4x - 3 & \text{với } x > 2. \end{cases}$

Tìm $\lim_{x \rightarrow 2^+} f(x)$, $\lim_{x \rightarrow 2^-} f(x)$ và $\lim_{x \rightarrow 2} f(x)$ (nếu có). {Bài tập 33/tr159_SGK ĐS11 NC}

Đầu tiên chúng ta nhập công thức của hàm số vào Maple:

> **f:=piecewise(x<=2,x^2-2*x+3,x>2,4*x-3);**

$$f := \begin{cases} x^2 - 2x + 3 & x \leq 2 \\ 4x - 3 & 2 < x \end{cases}$$

•Giới hạn bên phải của hàm số tại x=2:

> **limit(f,x=2,right);**

5

•Giới hạn bên trái của hàm số tại x=2:

> **limit(f,x=2,left);**

3

•Giới hạn của hàm số tại x=2:

> **limit(f,x=2);**

undefined

Hàm số không có giới hạn tại x=2.

Điều này dễ dàng nhận thấy vì $\lim_{x \rightarrow 2^+} f(x) = 5 \neq 3 = \lim_{x \rightarrow 2^-} f(x)$.

Qua đây chúng ta nhận thấy Maple có thể tính được giới hạn của những hàm số cho bởi nhiều công thức.

Maple cũng cho phép chúng ta thực hiện được các phép toán trên giới hạn (khi sử dụng gói lệnh **with(student)**).

Ví dụ 6: Hãy khảo sát các dòng lệnh sau:

> **with(student);**

> **L:=Limit(x^3-2*x^2+5/x,x=3/2);**

$$L := \lim_{x \rightarrow \frac{3}{2}} \left(x^3 - 2x^2 + \frac{5}{x} \right)$$

> **L:=expand(L);**

$$L := \left(\lim_{x \rightarrow \frac{3}{2}} x \right)^3 - 2 \left(\lim_{x \rightarrow \frac{3}{2}} x \right)^2 + \frac{5}{\lim_{x \rightarrow \frac{3}{2}} x}$$

> **value(%);**

$\frac{53}{24}$

Như vậy, Maple có thể phân tích giới hạn của một tổng thành tổng của các giới hạn thành phần. Và điều ngược lại Maple cũng thực hiện được.

Chúng ta hãy xem các dòng lệnh sau:

> **L1:=Limit((x+1)/(x^2-1),x=0)+Limit((sqrt(x+1)-1)/(x^2-3*x+2),x=0);**

$$L1 := \lim_{x \rightarrow 0} \left(\frac{x+1}{x^2-1} \right) + \lim_{x \rightarrow 0} \left(\frac{\sqrt{x+1}-1}{x^2-3x+2} \right)$$

> L1:=combine(L1);

$$L1 := \lim_{x \rightarrow 0} \left(\frac{x+1}{x^2-1} + \frac{\sqrt{x+1}-1}{x^2-3x+2} \right)$$

> value(%);

-1

Để kết hợp tổng các giới hạn chúng ta dùng hàm combine.

Ngoài ra còn có thể dùng gói lệnh “with(Student:-Calculus1):” để thực hiện các phép toán khi tính giới hạn một hàm số.

• **Quy tắc đưa hằng số ra khỏi dấu giới hạn trong phép nhân.**

Cú pháp: > Rule[**c***](Limit(**k*f(x)**),**x=a**));

Ví dụ:

> restart;

> with(Student:-Calculus1):

> Rule[**c***](Limit(sqrt(2)*sin(7*x), x=Pi/2));

$$\lim_{x \rightarrow \frac{1}{2}\pi} (\sqrt{2} \sin(7x)) = \sqrt{2} \lim_{x \rightarrow \frac{1}{2}\pi} \sin(7x)$$

> value(%);

$$\lim_{x \rightarrow \frac{1}{2}\pi} (\sqrt{2} \sin(7x)) = -\sqrt{2}$$

• **Quy tắc đổi biến.**

Cú pháp: > Rule[change, u=g(x)](Limit(**k*f(x)**),**x=a**));

Ví dụ: Khi tính giới hạn $\lim_{x \rightarrow +\infty} x \sin \frac{1}{x}$ chúng ta đổi biến $u = \frac{1}{x}$.

Maple sẽ biến đổi như sau:

> Rule[change, u=1/x](Limit(x*sin(1/x), x=+infinity));

$$\lim_{x \rightarrow \infty} \left(x \sin \left(\frac{1}{x} \right) \right) = \lim_{u \rightarrow 0^+} \left(\frac{\sin(u)}{u} \right)$$

II. Nguyên hàm và tích phân xác định.

1. Nguyên hàm của hàm số: $\int f(x)dx$.

Cú pháp: > Int(**f(x)**),**x**);

(Để hiển thị biểu thức nguyên hàm)

> int(**f(x)**),**x**);

(Để cho kết quả)

Ví dụ:

Để hiển thị biểu thức $\int \left(x^2 - 2x + \frac{1}{x} \right) dx$ chúng ta dùng lệnh thứ nhất:

> Int(x^2-2*x+1/x,x);

$$\int \left(x^2 - 2x + \frac{1}{x} \right) dx$$

Để hiển thị kết quả chúng ta dùng lệnh thứ hai (chữ i đầu trong từ khoá int là chữ thường):

> int(x^2-2*x+1/x,x);

$$\frac{1}{3} x^3 - x^2 + \ln(x)$$

Nếu muốn hiển thị cả hai, chúng ta có thể kết hợp hai lệnh trên như sau:

> **restart;**

> **Int(x^2-2*x+1/x,x)=int(x^2-2*x+1/x,x)+C;**

$$\int \left(x^2 - 2x + \frac{1}{x} \right) dx = \frac{1}{3} x^3 - x^2 + \ln(x) + C$$

Ở đây chúng ta cộng thêm hằng số C để kết quả trực quan và chính xác hơn !

- Một điều cần lưu ý rằng: $f(x)$ có thể là một hàm số, cũng có thể là một biểu thức chứa biến.

Chẳng hạn, khi nhập vào Maple:

> **f:=sqrt(x)-3*x^2+ln(x);**

$$f := \sqrt{x} - 3x^2 + \ln(x)$$

Thì Maple hiểu đó là một biểu thức, và có thể tìm nguyên hàm của biểu thức f như sau:

> **Int(f,x)=int(f,x)+C;**

$$\int (\sqrt{x} - 3x^2 + \ln(x)) dx = \frac{2}{3} x^{3/2} - x^3 + x \ln(x) - x + C$$

Nhưng nếu nhập dưới dạng hàm số:

> **f:=x->sqrt(x)-3*x^2+ln(x);**

$$f := x \rightarrow \sqrt{x} - 3x^2 + \ln(x)$$

> **Int(f,x)=int(f,x)+C;**

$$\int f dx = f x + C$$

Như vậy, khi f là hàm số, nếu trong câu lệnh chúng ta chỉ dùng tên hàm số “f” thì Maple không cho ra kết quả cụ thể, đôi lúc sai !

Chúng ta phải nhập “f(x)” thay cho “f”, khi đó Maple mới hiểu :

> **Int(f(x),x)=int(f(x),x)+C;**

$$\int (\sqrt{x} - 3x^2 + \ln(x)) dx = \frac{2}{3} x^{3/2} - x^3 + x \ln(x) - x + C$$

Đó chính là điều cần lưu ý khi sử dụng Maple !

2. Tích phân xác định của hàm số f trên đoạn [a; b] : $\int_a^b f(x) dx$.

Cú pháp: > **Int(f(x),x=a..b);** (Để hiện biểu thức tích phân)

> **int(f(x),x=a..b);** (Để cho kết quả)

Trong đó f(x) là một biểu thức hoặc là một hàm số biến số x .

Ví dụ: (SGK ĐS> 12)

> **Int(1/(x*(x+1)),x=1/2..2)=int(1/(x*(x+1)),x=1/2..2);**

$$\int_{\frac{1}{2}}^2 \frac{1}{x(x+1)} dx = \ln(2)$$

Có thể dùng thủ tục sau cho gọn khi muốn hiển thị cả hai lệnh trên:

```
> tp:=proc(f,a,b)
print(Int(f,x=a..b)=int(f,x=a..b))
end proc;
```

Để sử dụng thủ tục này chúng ta chỉ cần nhập >tp(f,a,b); thì sẽ cho kết quả tích phân của biểu thức f trên đoạn [a; b] .

Ví dụ: Sau khi đã thiết lập thủ tục trên chúng ta tính tích phân $\int_0^{\ln 2} \frac{e^{2x+1}+1}{e^x} dx$ như sau:

```
> f:=(e^(2*x+1)+1)/e^x;
```

$$f := \frac{e^{2x+1} + 1}{e^x}$$

```
> tp(f,0,ln(2));
```

$$\int_0^{\ln(2)} \frac{e^{2x+1} + 1}{e^x} dx = \frac{-e + 1 - e^{-\ln(2)} + e^{\ln(2) + 1}}{\ln(e)}$$

Một điều cần lưu ý, nếu f là hàm số thì lệnh tính giá trị của tích phân của hàm f trên đoạn [a; b] có thể ngắn gọn hơn là > int(f, a..b); _không khai báo biến x, chỉ khai báo “f” thay vì “f(x)”.

Ví dụ:

```
> f:= x -> x^2*(1-x)^5
```

Warning, inserted missing semicolon at end of statement

$$f := x \rightarrow x^2 (1 - x)^5$$

Chúng ta nhập f ở dạng hàm số.

```
> Int(f(x), x=0..1)=int(f,0..1);
```

$$\int_0^1 x^2 (1 - x)^5 dx = \frac{1}{168}$$

Nhưng muốn hiển thị biểu thức tích phân (khi f là hàm số) chúng ta phải khai báo đầy đủ

```
>Int(f(x), x=a..b); . Chúng ta hãy xem câu lệnh trên !
```

Còn nếu nhập lệnh không đúng, Maple sẽ cho kết quả :

```
> Int(f,0..1);
```

$$\text{Int}(f, 0..1)$$

♣ Gói lệnh “with(IntegrationTools)” với tích phân.

1. Tính tích phân bằng phương pháp đổi biến số

Cú pháp > Change(V, x=g(u));

Trong đó V là biểu thức tích phân của một hàm số xác định trước.

Ví dụ: Tính $\int_0^1 \sqrt{1-x^2} dx$ (Hướng dẫn: đặt $x = \sin u$)

{Bài tập, SGK ĐS-GT 12}

Chúng ta sử dụng Maple để thực hiện các bước tính tích phân trên, như sau:

Đầu tiên, chúng ta nhập biểu thức tích phân đã cho:

```
> with(IntegrationTools):
```

```
> V:=Int(sqrt(1-x^2),x=0..1);
```

$$V := \int_0^1 \sqrt{1-x^2} dx$$

Đổi biến theo hướng dẫn, chúng ta được kết quả:

```
> V:=Change(V,x=sin(u));
```

$$V := \int_0^{\frac{1}{2}\pi} \cos(u)^2 du$$

Tiếp theo chúng ta có thể dùng lệnh “subs” để khai triển biểu thức dưới dấu tích phân:

```
> V:=subs(cos(u)^2=(1+cos(2*u))/2,V1);
```

$$V := \int_0^{\frac{1}{2}\pi} \left(\frac{1}{2} + \frac{1}{2} \cos(2u) \right) du$$

Đến đây, việc tính tích phân trên không còn phức tạp nữa.

Để hiển thị kết quả chúng ta dùng lệnh:

```
> V:=value(V);
```

$$V := \frac{1}{4} \pi$$